Practice of Answer Set Programming

Graph Problems in ASP (II)

Ira A. Fulton Schools of Engineering
Arizona State University

Objectives

Objective

Use ASP to solve graph-related problems

Vertex Cover

Definition

- A **vertex cover** of a graph is a set of vertices such that every edge of the graph has at least one endpoint in that set

- We would like to find a vertex cover of size n





Example

```
% in/1 is a set consisting of n vertices of G
{in(X) : vertex(X)} = n.

% covered/2 is the set of edges of G that have
% an endpoint in in/1
covered(X,Y) :- edge(X,Y), in(X).
covered(X,Y) :- edge(X,Y), in(Y).

% every edge of G is in covered/2
:- edge(X,Y), not covered(X,Y).
```

Graph

```
% Nodes
vertex(1..6).


% Edges
edge(1,(2;3;4)). edge(4,(1;2)).

edge(2,(4;5;6)). edge(5,(3;4;6)).

edge(3,(1;4;5)). edge(6,(2;3;5)).
```



Graph

```
vertex(a; b; c; d; e; f).

edge(a,b; b,c; c,a; d,f; f,e; e,d; a,d; f,c; b,e).
```



Hamiltonian Cycle

Introduction



Introduction





Graph

```
% Nodes
vertex(1..6).


% Edges
edge(1,(2;3;4)). edge(4,(1;2)).

edge(2,(4;5;6)). edge(5,(3;4;6)).

edge(3,(1;4;5)). edge(6,(2;3;5)).
```



Hamiltonian Cycle

A Hamiltonian cycle in a graph G is a subgraph C of G that has the same set V of vertices as G and satisfies two conditions:

1. Every vertex has only one incoming edge and only one outgoing edge in the cycle.
2. Every vertex is reachable in C from some fixed vertex, say 1.

Hamiltonian Cycle in ASP

```
% in(X,Y) is the set of edges included in the cycle and
% satisfies Condition 1.
{in(X,Y) : edge(X,Y)} = 1 :- vertex(X).
{in(X,Y) : edge(X,Y)} = 1 :- vertex(Y).

% Define reachability recursively
reachable(X) :- in(1,X).
reachable(Y) :- in(X,Y), reachable(X).

% Satisfies Condition 2
:- not reachable(X), vertex(X).

% Display
#show in/2.
```

Wrap-Up