Practice of Answer Set Programming

Representing Functions in ASP

Ira A. Fulton Schools of Engineering
Arizona State University

Objectives

Objective

Represent functions, 1-1 functions, onto functions, 1-1 correspondence in ASP

Function

A **function** from a set A to a set B is a relation f from A X B such that for every element x in A, there is exactly one element y in B such that (x, y) is in f.

Q: A = {1,2,3}, B = {a,b}. How many functions are there?

(a) 1 (b) 8 (c) 16 (d) 27

$|A| = m$

$|B| = n$

$\underbrace{n \times n \cdots n}_{m} = n^m$



Representing Functions in ASP

A **function** from a set A to a set B is a relation f from A X B such that for every element x in A, there is exactly one element y in B such that (x,y) is in f.

```
% function.lp


domain(1;2;3).
codomain(a;b).


{f(X,Y): codomain(Y)} = 1 :- domain(X).

#show f/2.
```

{f(1,Y) : codomain(Y)} = 1.

⇓

{f(1,a) ; f(1,b)} = 1.

One-to-One (1-1) (a.k.a. Injective)

1-1: A function f from A to B is called one-to-one (1-1) if for any elements (x, y) in f and $(x1, y)$ in f, it is the case that $x = x1$.



Q: A = {1,2,3}, B = {a,b}. How many one-to-one functions? 0

Q: A = {1,2,3}, B = {a,b,c,d,e}. How many one-to-one functions?

1 2 3

$5 \times 4 \times 3 = 60$

Representing 1-1 Functions in ASP

1-1 function: A function f from A to B is called one-to-one (1-1) if for any elements (x, y) in f and $(x1, y)$ in f, it is the case that $x = x1$.

```
% one-to-one-function.lp

domain(1;2;3).
codomain(a;b;c;d;e).

{f(X,Y): codomain(Y)}=1 :- domain(X).
X=X1 :- f(X,Y), f(X1,Y).

#show f/2.
```



Onto (a.k.a. Surjective)

Onto: A function f from A to B is called **onto** if for each element y in B, there is an element x in A such that $(x, y) \in f$

Q: How many onto functions are there when

- A = {1,2,3}, B = {a}. 1
- A = {1,2,3}, B = {a,b}. $2^3 - 2 = 6$
- A = {1,2,3}, B = {a,b,c}. $3^3 - 3 \cdot 6 - 3 \cdot 1 = 6$

$$= 3^3 - \binom{3}{2} \cdot 6 - \binom{3}{1} \cdot 1$$



Onto (a.k.a. Surjective)

Onto: A function f from A to B is called **onto** if for each element y in B, there is an element x in A such that $(x, y) \in f$

Q: How many onto functions are there when

- A = {1,2,3}, B = {a,b,c,d}. 0

- A = {1,2,3}, B = {a,b,c,d,e}. 0

Representing Onto Functions in ASP

Onto: A function f from A to B is called **onto** if for each element y in B, there is an element x in A such that $(x, y) \in f$

```
% onto-function.lp

domain(1;2;3).
codomain(a;b).

{f(X,Y): codomain(Y)} = 1 :- domain(X).
:- {f(X,Y): domain(X)} = 0, codomain(Y).

#show f/2.
```

1-1 Correspondence (a.k.a. Bijective)

1-1 correspondence: A function is called a **1-1 correspondence** if it is both 1-1 and onto

Q: How many 1-1 correspondences when

- A = {1,2,3}, B = {a,b}. 0
- A = {1,2,3}, B = {a,b,c}. $3 \times 2 \times 1 = 6$
- A = {1,2,3}, B = {a,b,c,d}. 0

Representing 1-1 Correspondence in ASP

1-1 correspondence: A function is called a **1-1 correspondence** if it is both 1-1 and onto

```
% one-to-one-correspondence.lp

domain(1;2;3).
codomain(a;b;c).

{f(X,Y): codomain(Y)} = 1 :- domain(X).
X = X1 :- f(X,Y), f(X1,Y).
:- {f(X,Y): domain(X)} = 0, codomain(Y).

#show f/2.
```



Wrap-Up