Practice of Answer Set Programming

Combinatorial Search

Ira A. Fulton Schools of Engineering
Arizona State University

Objectives

Objective

Apply Generate-Test and Function Representation in ASP to combinatorial search problems

Seating Arrangements

Seating Arrangements

There are n chairs around a table. Choose a chair for each of n guests so that guests who like each other sit next to each other, and guests who don ’t like each other sit at least one chair away.

Sample input:

```
#const n=6.

  like(1,2; 3,4).

  dislike(2,3; 1,3).
```



Seating Arrangements in ASP

```
% at(G,C) means that guest G is assigned chair C.
% each guest is assigned a chair.
{at(G,1..n)} = 1 :- G = 1..n.

% different guests are assigned different chairs.
G1 = G2 :- at(G1,C), at(G2,C).

% adj(X,Y) iff chair X is adjacent to chair Y.
adj(X,Y) :- X = 1..n, Y = 1..n, |X-Y| = 1.
adj(1,n; n,1).

% guests who like each other sit next to each other.
:- like(G1,G2), at(G1,C1), at(G2,C2), not adj(C1,C2).

% guests who don't like each other don't sit to each other.
:- dislike(G1,G2), at(G1,C1), at(G2,C2), adj(C1,C2).

#show at/2.
```



Logic Puzzle

Logic Puzzle

Each of four men owns a different species of exotic pet.

1. Mr. Engels (whose pet is named Sparky), Abner and Mr. Foster all belong to a club for owners of unusual pets.
2. The iguana is not owned by either Chuck or Duane.
3. Neither the jackal nor the king cobra is owned by Mr. Foster.
4. The llama does not belong to Duane (whose pet is named Waggles).
5. Abner, who does not own the king cobra, is not Mr. Gunter.
6. Bruce and Mr. Foster are neighbors
7. Mr. Halevy is afraid of iguanas.

Q: Who owns the jackal?

Logic Puzzle in ASP (I)

```
first_name(abner; bruce; chuck; duane).
last_name(engels; foster; gunter; halevy).
pet(iguana; jackal; king_cobra; llama).

% a unique last name and unique pet species are chosen for
each first name.
{full_name(F,L) : last_name(L)} = 1 :- first_name(F).
{owns(F,P) : pet(P)} = 1 :- first_name(F).

% the chosen names and pets are pairwise
F1 = F2 :- full_name(F1,L), full_name(F2,L).
F1 = F2 :- owns(F1,P), owns(F2,P).
```

Logic Puzzle in ASP (II)

```
% Abner's last name is neither Engels nor Foster.
:- full_name(abner,engels).
:- full_name(abner,foster).

% iguana belongs neither to Chuck nor to Duane
:- owns(chuck,iguana).
:- owns(duane,iguana).

% Mr.Foster owns neither jackal nor king cobra.
:- full_name(X,foster), owns(X,jackal).
:- full_name(X,foster), owns(X,king_cobra).

% Duane's pet is not llama
:- owns(duane,llama).

% Duane's last name is not Engels.
:- full_name(duane,engels).
```

Logic Puzzle in ASP (III)

```
% Abner's pet is not king cobra.
:- owns(abner,king_cobra).

% Abner's last name is not Gunter
:- full_name(abner,gunter).

% Bruce's last name is not Foster.
:- full_name(bruce,foster).

% Mr. Halevy's pet is not iguana.
:- full_name(X,halevy), owns(X,iguana).

answer(X,Y) :- full_name(X,Y), owns(X,jackal).

#show answer/2.
```



Wrap-Up